
* REPLICATION FILE FOR THE PAPER: Foxes and Hedgehogs in Legislatures: Why do some MPs become policy specialists and others generalists?
* Javier Martíınez–Cantó (University of Konstanz), Christian Breunig (University of Konstanz), and Laura Chaqués-Bonafont (University of Barcelona)
* December 8, 2022

* All the analyses were done using Stata (version 14).
* Note: To run the replication file smoothly, create a folder called "results" in your working directory. Moreover, the following user-written stata programs are necessary: coefplot, ciplot, estout.

clear all
use "replication_data.dta"
set scheme plotplain

* FIGURE 1 MAIN TEXT
***************************
**** DESCRIPTIVE PLOTS ****
***************************
{
*** Histogram
gen hhi2 = hhi
replace hhi2 = -1 if hhi2==.
label define hhi2 -1 "Non-speaker"
label value hhi2 hhi2

* Total Distribution
hist hhi2 	, percent ylabel(0(10)40) xlabel(-1 0 .25 .5 .75 1, valuelabel) xscale(range(-1.25 1.25)) ///
								xtitle("Herfindahl–Hirschman index (0-1)", size(small)) title("(a) Issue specialization in Germany", size(small)) ytitle("Percent", size(small)) ///
								name(distribution, replace) fysize(200) fxsize(200)

* Distribution over time
ciplot hhi asked, by(lp2) horizontal xlabel(0(.2)1) name(timechange, replace) total inclusive  ///
	rcapopts(recast(rspike) lcolor(red) lwidth(none)) ///
	ytitle("") title("(b) Temporal differences", position(12) size(small)) note("")	xtitle("Herfindahl–Hirschman index (0-1) and Proportion of MPs participating in questioning", size(small)) /// 
	legend(position(6) rows(1) label(2 "Average HHI") label(3 "Proportion of MPs participating in questioning") size(vsmall)) ///
	fysize(200) fxsize(200)


graph combine distribution timechange, rows(2) ysize(20) xsize(20)
graph export "results/histogram_germany.png", replace width(1800)

*
}
*

*************************************
****** GLOBALS: VARIABLES SETS ******
*************************************
{
*** Demand factors
global demand governing_party2 parliamentary_leadership std_seats_party std_number_committees2 
global demand_ i.governing_party2 i.parliamentary_leadership c.std_seats_party c.std_number_committees2 

*** Individual Controls
global ind_controls  gender std_seniority std_vulnerability smd committees_spokesperson std_plenary_q 
global ind_controls_ i.gender c.std_seniority c.std_vulnerability i.smd i.committees_spokesperson c.std_plenary_q 

*** Party Controls
global party_controls std_nicheness_meyer_std std_rile
global party_controls_ c.std_nicheness_meyer_std c.std_rile 

*** Selection into Sample
global selection governing_party2 parliamentary_leadership std_seats_party gender std_number_committees2 smd
global selection_ i.governing_party2 i.parliamentary_leadership c.std_seats_party i.gender c.std_number_committees2 i.smd

*** Party FE
global partyfe i.party 
global partyfe_ i.party 

*** LP FE
global lpfe i.lp
global lpfe_ i.lp

*** I remove missing observations to ensure the consistency of the sample
*drop if isei == .
drop if vulnerability == .
drop if age == .
drop if seniority == .
drop if number_committees2_share == .
drop if seats_party_log == .
drop if nicheness_meyer_std == .
drop if rile == .

}
*

* APPENDIX A1
******************************
*** Descriptive statistics ***
******************************
{
*** Descriptive statistics

tabulate party, generate(parties_)
foreach var of varlist parties_* {
    local variable_label : variable label `var'
    local variable_label : subinstr local variable_label "party==" ""
    label variable `var' "`variable_label'"
}
*
tabulate lp, generate(lp_)
foreach var of varlist lp_* {
    local variable_label : variable label `var'
    local variable_label : subinstr local variable_label "lp==" ""
    label variable `var' "`variable_label'"
}
*

eststo descriptives_germany: quietly estpost summarize ///
    hhi asked $demand $ind_controls $party_controls lp_* parties_* share_biggest_topic 

esttab descriptives_germany, ///
cells("mean( fmt(3)) sd( fmt(3)) min( fmt(3)) max( fmt(3))") ///
label compress // mtitle("") 

esttab descriptives_germany using "results/descriptive_germany.tex", replace ///
cells("mean( fmt(3)) sd( fmt(3)) min( fmt(3)) max( fmt(3))") booktabs alignment(D{.}{.}{-1}) ///
label compress title("Descriptive statistics") //

*
}
*

* MAIN RESULTS - APPENDIX A2
************************************************************
*** Models - Using Herfindahl–Hirschman Index as outcome *** 
************************************************************
{

*** Stepwise model
heckman hhi 	$partyfe $lpfe																							, select(asked = $selection ) vce(cluster name_id) 
eststo germany1
heckman hhi $demand $partyfe $lpfe																						, select(asked = $selection ) vce(cluster name_id) 
eststo germany2
heckman hhi $ind_controls $party_controls $partyfe $lpfe																	, select(asked = $selection ) vce(cluster name_id) 
eststo germany3
heckman hhi $demand $ind_controls $party_controls $partyfe $lpfe															, select(asked = $selection ) vce(cluster name_id) 
eststo germany4


*** COMPARISON 

esttab germany* , ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level.") order($demand std_number_committees2 $supply $ind_controls) 

esttab germany* using "results/mainresults_germany.rtf", ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level. Standarized variables") order($demand std_number_committees2 $supply $ind_controls) mtitle("Model 1" "Model 2" "Model 3" "Model 4") title("Determinants of MPs’ agenda specialization") 

esttab germany* using "results/mainresults_germany.tex", ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level. Standarized variables") order($demand std_number_committees2 $supply $ind_controls) mtitle("Model 1" "Model 2" "Model 3" "Model 4") title("Determinants of MPs’ agenda specialization") ///
	booktabs alignment(D{.}{.}{-1}) longtable //
	
*
}
*

* FIGURE 2 MAIN TEXT
****************	
*** Coefplot ***
****************
{
*center $ind_controls $party_controls $demand germany $controls, inplace standardize addtolabel("")

heckman hhi $demand $ind_controls $party_controls $partyfe $lpfe															, select(asked = $selection ) vce(cluster name_id) 
eststo coefplot_germany

coefplot 		(coefplot_germany, label("") msymbol(circle) mcolor(black*.99) ciopts(lcolor(black*.99))), bylabel("{bf:Participation in questioning (Eq. 1)}") keep(asked:*)   ///
			||	(coefplot_germany, label("") msymbol(circle) mcolor(black*.99) ciopts(lcolor(black*.99))), bylabel("{bf:Herfindahl–Hirschman index (Eq. 2)}") keep(hhi:*)   ///
			||	, drop(_cons *.party *.lp) xline(0) order($demand std_number_committees2 $supply $ind_controls)  /// 
			headings(governing_party2 = "{bf:Party factors:}" gender = "{bf:Adjustment variables:}") ///			
			scheme(plotplain) xtitle("Standardized Coefficients", size(small)) legend( position(6) rows(1)) ///  
			byopts(xrescale   cols(2)) subtitle(,bcolor(white)) // 

graph export "results/coefplot_bn_germany.png", replace width(1800)


}
*

* FIGURE 3 MAIN TEXT
***************
*** Margins ***
***************
{
*** Government status 
heckman hhi $demand $ind_controls $party_controls $partyfe $lpfe															, select(asked = $selection ) vce(cluster name_id) 
margins , at(governing_party2=(0 1)) vsquish atmeans predict(ycond)
marginsplot, name(file1, replace) recast(scatter) recastci(rspike) ytitle("Herfindahl–Hirschman index", size(small)) scheme(plotplain) title("(a) Government status") yscale(range(0(.2)1)) ylabel(0(.2)1) xscale(range(-0.2 1.2)) xlabel(0 1) xlabel(,labsize(small)) xtitle("") plot1opts(lcolor(black) mcolor(black) msymbol(circle))

*** Leadership position 
heckman hhi $demand $ind_controls $party_controls $partyfe $lpfe															, select(asked = $selection ) vce(cluster name_id) 
margins , at(parliamentary_leadership=(0 1)) vsquish atmeans predict(ycond)
marginsplot, name(file2, replace) recast(scatter) recastci(rspike) ytitle("Herfindahl–Hirschman index", size(small)) scheme(plotplain) title("(b) Leadership position") yscale(range(0(.2)1)) ylabel(0(.2)1) xscale(range(-0.2 1.2)) xlabel(0 1) xlabel(,labsize(small)) xtitle("") plot1opts(lcolor(black) mcolor(black) msymbol(circle))

*** Party Size 
heckman hhi $demand $ind_controls $party_controls $partyfe $lpfe															, select(asked = $selection ) vce(cluster name_id) 
margins , at(std_seats_party=(-2(.25)1)) vsquish atmeans predict(ycond)
marginsplot, name(file3, replace) recast(scatter) recastci(rspike) ytitle("Herfindahl–Hirschman index", size(small)) scheme(plotplain) title("(c) Party size") yscale(range(0(.2)1)) ylabel(0(.2)1) xscale(range(-2(.5)1)) xlabel(-2(1)1) xlabel(,labsize(small)) xtitle("") plot1opts(lcolor(black) mcolor(black) msymbol(circle))

*** Committee 
heckman hhi $demand $ind_controls $party_controls $partyfe $lpfe															, select(asked = $selection ) vce(cluster name_id) 
margins , at(std_number_committees2=(-2(.5)2)) vsquish atmeans predict(ycond)
marginsplot, name(file4, replace) recast(scatter) recastci(rspike) ytitle("Herfindahl–Hirschman index", size(small)) scheme(plotplain) title("(d) Committee participation") yscale(range(0(.2)1)) ylabel(0(.2)1) xscale(range(-2(.5)2)) xlabel(-2(1)2) xlabel(,labsize(small)) xtitle("") plot1opts(lcolor(black) mcolor(black) msymbol(circle))

*** Graph combine
graph combine file1 file2 file3 file4, rows(2) xsize(8) ysize(8)
graph export "results/margins_together_germany.png", replace width(1800)

*
}
*

* APPENDIX A3
********************************************
*** Interaction: Government X Party Size ***
********************************************
{
heckman hhi $demand_ $ind_controls_ $party_controls_ $partyfe_ $lpfe_ 	i.governing_party2#c.std_seats_party , select(asked = $selection_ ) vce(cluster name_id) 
eststo interaction
margins governing_party2, at(std_seats_party=(-1(.2)1)) vsquish atmeans 
marginsplot, subtitle("") recast(scatter) recastci(rspike) ytitle("Pr(Herfindahl–Hirschman index)", size(small)) scheme(plotplain) title("") yscale(range(0(.2)1)) ylabel(-0.2(.2)1) xscale(range(-1(.1)1)) xlabel(-1(1)1) xlabel(,labsize(small)) xtitle("") plot1opts(lcolor(black) mcolor(black) msymbol(circle))
graph export "results/interaction_results_germany.png", replace

esttab interaction* , ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level.") 

esttab interaction* using "results/interaction_results_germany.rtf", ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level. Standarized variables") mtitle("Model 1" "Model 2" "Model 3" "Model 4") title("Interaction between government status and PPG size") //  

esttab interaction* using "results/interaction_results_germany.tex", ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level. Standarized variables") mtitle("Model 1" "Model 2" "Model 3" "Model 4") title("Interaction between government status and PPG size") /// 
	booktabs alignment(D{.}{.}{-1}) longtable 

	
}
*

* APPENDIX D	
********************************************************************
*** ROBUSTNESS - Using the share of the biggest topic as outcome *** 
********************************************************************
{

*** Stepwise model
heckman share_biggest_topic 	$partyfe $lpfe																							, select(asked = $selection ) vce(cluster name_id) 
eststo share1
heckman share_biggest_topic $demand $partyfe $lpfe																						, select(asked = $selection ) vce(cluster name_id) 
eststo share2
heckman share_biggest_topic $ind_controls $party_controls $partyfe $lpfe																	, select(asked = $selection ) vce(cluster name_id) 
eststo share3
heckman share_biggest_topic $demand $ind_controls $party_controls $partyfe $lpfe															, select(asked = $selection ) vce(cluster name_id) 
eststo share4


*** COMPARISON 

esttab share* , ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level.") order($demand std_number_committees2 $supply $ind_controls) //

esttab share* using "results/share_biggest_topic_germany.rtf", ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level. Standarized variables") order($demand std_number_committees2 $supply $ind_controls) mtitle("Model 1" "Model 2" "Model 3" "Model 4") title("Determinants of MPs’ agenda specialization (using top issue proportion as outcome)") //  

esttab share* using "results/share_biggest_topic_germany.tex", ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level. Standarized variables") order($demand std_number_committees2 $supply $ind_controls) mtitle("Model 1" "Model 2" "Model 3" "Model 4") title("Determinants of MPs’ agenda specialization (using top issue proportion as outcome)") ///
	booktabs alignment(D{.}{.}{-1}) longtable ///
	
*
}
*

* APPENDIX E
************************************************
*** ROBUSTNESS - Types of electoral district *** 
************************************************
{
heckman hhi 	$partyfe $lpfe i.urban_rural																						, select(asked = $selection ) vce(cluster name_id) 
eststo urbanrural1
heckman hhi $demand $partyfe $lpfe i.urban_rural																					, select(asked = $selection ) vce(cluster name_id) 
eststo urbanrural2
heckman hhi $ind_controls $party_controls $partyfe $lpfe i.urban_rural																	, select(asked = $selection ) vce(cluster name_id) 
eststo urbanrural3
heckman hhi $demand $ind_controls $party_controls $partyfe $lpfe i.urban_rural															, select(asked = $selection ) vce(cluster name_id) 
eststo urbanrural4
heckman hhi $demand $ind_controls $party_controls $partyfe $lpfe i.urban_rural##c.safety_district										, select(asked = $selection ) vce(cluster name_id) 
eststo urbanrural5
heckman hhi $demand $ind_controls $party_controls $partyfe $lpfe i.urban_rural	if smd==1														, select(asked = $selection ) vce(cluster name_id) 
eststo urbanrural6


*** COMPARISON 

esttab urbanrural* , ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level.") order($demand std_number_committees2 $supply $ind_controls) //

esttab urbanrural* using "results/urbanrural.rtf", ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level. Standarized variables") order($demand std_number_committees2 $supply $ind_controls) mtitle("Model 1" "Model 2" "Model 3" "Model 4") title("Determinants of MPs’ agenda specialization (including type of single-member district)") //  

esttab urbanrural* using "results/urbanrural.tex", ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Clustered standard errors at the MP level. Standarized variables") order($demand std_number_committees2 $supply $ind_controls) mtitle("Model 1" "Model 2" "Model 3" "Model 4") title("Determinants of MPs’ agenda specialization (including type of single-member district)") ///
	booktabs alignment(D{.}{.}{-1}) longtable ///
	
*
}
*	

* APPENDIX C
*****************************
*** ROBUSTNESS - Matching *** 
*****************************
{
* Create new dummy variable for the continious variable
egen mean_std_seats_party = mean(std_seats_party)
gen dummy_std_seats_party = 0
replace dummy_std_seats_party = 1 if std_seats_party > mean_std_seats_party
tab dummy_std_seats_party

xtile pct_std_seats_party = std_seats_party, nquantiles(4)
replace pct_std_seats_party = . if pct_std_seats_party==2
replace pct_std_seats_party = . if pct_std_seats_party==3
replace pct_std_seats_party = 0 if pct_std_seats_party==1
replace pct_std_seats_party = 1 if pct_std_seats_party==4

egen mean_std_number_committees2 = mean(std_number_committees2)
gen dummy_std_number_committees2 = 0
replace dummy_std_number_committees2 = 1 if std_number_committees2 > mean_std_number_committees2
tab dummy_std_number_committees2

xtile pct_std_number_committees2 = std_number_committees2, nquantiles(4)
replace pct_std_number_committees2 = . if pct_std_number_committees2==2
replace pct_std_number_committees2 = . if pct_std_number_committees2==3
replace pct_std_number_committees2 = 0 if pct_std_number_committees2==1
replace pct_std_number_committees2 = 1 if pct_std_number_committees2==4

*** Exact matching
* Government
teffects nnmatch (hhi /*governing_party2*/ parliamentary_leadership std_seats_party std_number_committees2 gender std_seniority std_vulnerability smd committees_spokesperson std_plenary_q) (governing_party2) , dmvariables
eststo matching1
* Parliamentary leadership
teffects nnmatch (hhi governing_party2 /*parliamentary_leadership*/ std_seats_party std_number_committees2 gender std_seniority std_vulnerability smd committees_spokesperson std_plenary_q) (parliamentary_leadership) , dmvariables
eststo matching2
* PPG size
teffects nnmatch (hhi governing_party2 parliamentary_leadership /*std_seats_party*/ std_number_committees2 gender std_seniority std_vulnerability smd committees_spokesperson std_plenary_q) (dummy_std_seats_party) , dmvariables
eststo matching3
* Committees
teffects nnmatch (hhi governing_party2 parliamentary_leadership std_seats_party /*std_number_committees2*/ gender std_seniority std_vulnerability smd committees_spokesperson std_plenary_q) (dummy_std_number_committees2) , dmvariables
eststo matching4
* PPG size 2
teffects nnmatch (hhi governing_party2 parliamentary_leadership /*std_seats_party*/ std_number_committees2 gender std_seniority std_vulnerability smd committees_spokesperson std_plenary_q) (pct_std_seats_party) , dmvariables
eststo matching5
* Committees 2
teffects nnmatch (hhi governing_party2 parliamentary_leadership std_seats_party /*std_number_committees2*/ gender std_seniority std_vulnerability smd committees_spokesperson std_plenary_q) (pct_std_number_committees2) , dmvariables
eststo matching6

esttab matching* , ///
	coeflabel(r1vs0.governing_party2 "ATE Party in government" r1vs0.parliamentary_leadership "ATE PPG leadership" r1vs0.dummy_std_seats_party "ATE Party Size (Def. 1)" r1vs0.dummy_std_number_committees2 "ATE Committee Participation (Def. 1)" r1vs0.pct_std_seats_party "ATE Party Size (Def. 2)" r1vs0.pct_std_number_committees2 "ATE Committee Participation (Def. 2)") ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  /// 
	addnotes("Nearest-neighbor matching using the Mahalanobis distance.""MPs matched on the rest of party factors and all our adjustment variables as in the main results") title("Determinants of MPs’ agenda specialization") mtitle("Government status" "Leadership position" "Party Size" "Committee participation" "Party Size 2" "Committee participation 2") //

esttab matching* using "results/matching_germany.rtf", ///
	coeflabel(r1vs0.governing_party2 "ATE Party in government" r1vs0.parliamentary_leadership "ATE PPG leadership" r1vs0.dummy_std_seats_party "ATE Party Size (Def. 1)" r1vs0.dummy_std_number_committees2 "ATE Committee Participation (Def. 1)" r1vs0.pct_std_seats_party "ATE Party Size (Def. 2)" r1vs0.pct_std_number_committees2 "ATE Committee Participation (Def. 2)") ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  ///
	addnotes("Nearest-neighbor matching using the Mahalanobis distance.""MPs matched on the rest of party factors and all our adjustment variables as in the main results.") title("Determinants of MPs’ agenda specialization") mtitle("Model 1" "Model 2" "Model 3" "Model 4" "Model 5" "Model 6")  //

esttab matching* using "results/matching_germany.tex", ///
	coeflabel(r1vs0.governing_party2 "ATE Party in government" r1vs0.parliamentary_leadership "ATE PPG leadership" r1vs0.dummy_std_seats_party "ATE Party Size (Def. 1)" r1vs0.dummy_std_number_committees2 "ATE Committee Participation (Def. 1)" r1vs0.pct_std_seats_party "ATE Party Size (Def. 2)" r1vs0.pct_std_number_committees2 "ATE Committee Participation (Def. 2)") ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  ///
	addnotes("Nearest-neighbor matching using the Mahalanobis distance.""MPs matched on the rest of party factors and all our adjustment variables as in the main results.") title("Determinants of MPs’ agenda specialization") mtitle("Model 1" "Model 2" "Model 3" "Model 4" "Model 5" "Model 6")  ///
	booktabs alignment(D{.}{.}{-1}) longtable //

*
}
*

* APPENDIX B
*********************************************
*** ROBUSTNESS - Fixed and Random Effects *** 
*********************************************
{
* PANEL DATA SETTING
xtset name_id lp, generic

*** Stepwise model Fixed Effects
xtreg hhi $demand $partyfe $lpfe																						, fe
eststo multilevel1
xtreg hhi $ind_controls $party_controls $partyfe $lpfe																	, fe
eststo multilevel2
xtreg hhi $demand $ind_controls $party_controls $partyfe $lpfe															, fe
eststo multilevel3

*** Stepwise model Random Effects
xtreg hhi $demand $partyfe $lpfe																						, re
eststo multilevel4
xtreg hhi $ind_controls $party_controls $partyfe $lpfe																	, re
eststo multilevel5
xtreg hhi $demand $ind_controls $party_controls $partyfe $lpfe															, re
eststo multilevel6

esttab multilevel* , ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Random effects multilevel model, MP-Legislative Period observations grouped in MPs.") order($demand std_number_committees2 $supply $ind_controls) mtitle("FE 1" "FE 2" "FE 3" "RE 1" "RE 2" "RE 3") title("Determinants of MPs’ agenda specialization") //

esttab multilevel* using "results/multilevel_germany.rtf", ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Random effects multilevel model, MP-Legislative Period observations grouped in MPs.") order($demand std_number_committees2 $supply $ind_controls) mtitle("FE 1" "FE 2" "FE 3" "RE 1" "RE 2" "RE 3") title("Determinants of MPs’ agenda specialization") //

esttab multilevel* using "results/multilevel_germany.tex", ///
	label mtitles nonumbers replace compress se noomitted star(* 0.05 ** 0.01 *** 0.001) dropped("Ref.")  indicate("Party FE = *.party" "LP FE = *.lp") /// 
	addnotes("Random effects multilevel model, MP-Legislative Period observations grouped in MPs.") order($demand std_number_committees2 $supply $ind_controls) mtitle("FE 1" "FE 2" "FE 3" "RE 1" "RE 2" "RE 3") title("Determinants of MPs’ agenda specialization") ///
	booktabs alignment(D{.}{.}{-1}) longtable //
	
*
}
*

